

NOTE

Insect Label Production Using a Personal Computer

A personal computer and an "inexpensive" dot matrix printer make it possible to prepare small runs of insect and other labels in a very efficient manner. The system I use is a CROMEMCO Z2D[®] computer with a Texas Instruments 820[®] printer.

The TI820[®] compressed print option fits 9 characters on the width of a 0.5 inch wide label; three lines of print occupy 0.5 inch. I have found these labels to be of the proper dimension for use of point-mounted Coleoptera. The system is fast since the printer runs at 150 characters per second.

The print quality of the label is about that of a good typewriter except that each letter is formed from a series of 7 vertical dots instead of being fully formed as with a typewriter. The listing of the program will indicate the general legibility of the print. The TI820[®] uses dual tractors with pin feed so that only computer forms can be printed. I use a continuous computer form which is card stock containing 3 × 5 inch cards such as is used by libraries in cataloging their holdings.

The computer program presented here uses a BASIC interpreter and will run on most personal computers (with BASIC) except that the procedure to fill a string is unique to the CROMEMCO 16K BASIC[®] which was used to develop and run the program.

The program assumes that input from the user comes from the keyboard of the printer. A CRT terminal keyboard could be used for input but then statements 340 through 360 of the program must be changed so that the output is written to the printer instead of to the CRT.

The program produces eight (8) 10-spaced labels as a group with a blank space between each label. The number of lines per label is here limited to three but the number of lines and the width of the label can be changed by the user upon modifying the program.

The user first specifies how many groups of 8 labels is wanted by supplying a "REPLICATION FACTOR." For instance, if he wants 80 labels he would enter "10." Next, the user supplies information for each line of the label following the prompt from the program. A small label of one or two lines can be prepared by entering a "*" as indicated by the program. After producing the labels wanted the program loops back to the beginning and asks for the next REPLICATION FACTOR. Any negative number, i.e. -1, will stop the program.

A listing of the program and an example of its use is shown in Fig. 1.

```

10  Rem PROGRAM TO PRODUCE INSECT LABELS
20  Rem AUTHOR: D. G. Kissinger
30  Rem
40  Rem PRODUCES 8 LABELS ON A LINE EACH LABEL 9 CHARACTERS WIDE AND
50  Rem SEPARATED FROM ADJACENT LABELS BY A SPACE.
60  Rem
70  Rem SET UP TO PRODUCE LABELS WITH AT MOST 3 LINES
80  Rem
90  Dim A$(79),B$(79),C$(79),N$(9),Z$(9) : Rem DIMENSION STRINGS
100 Z$(-1)=" "+Z$(-1) : Rem SET Z$ TO ALL BLANKS
110 Rem BEGIN MAIN LOOP OF PROGRAM - STATEMENTS 120 THROUGH 400
120 Input"WHAT REPLICATION FACTOR (EXIT WITH -1)  ",G
130 If G<0 Then Stop
140 Input"DATA FOR LINE 1  ",N$(0,8) : Rem ENTER UP TO 9 CHARACTERS
150 N$(Len(N$))=Z$ : Rem SET UNUSED PORTION OF LINE TO BLANKS
160 A$(-1)=N$+A$(-1) : Rem PRODUCE 8 COPIES OF FIRST LINE
170 L=1 : Rem THERE IS AT LEAST ONE LINE IN THE LABEL
180 N$="" : Rem EMPTY THIS STRING
190 Input"DATA FOR LINE 2 (END LABEL WITH *)  ",N$(0,8)
200 If N$(0,0)="*" Then 310
210 L=2 : Rem THERE ARE TWO LINES IN THE LABEL
220 N$(Len(N$))=Z$
230 B$(-1)=N$+B$(-1)
240 N$="" : Rem EMPTY THIS STRING
250 Input"DATA FOR LINE 3 (END LABEL WITH *)  ",N$(0,8)
260 If N$(0,0)="*" Then 310
270 L=3 : Rem LABEL CONTAINS 3 LINES
280 N$(Len(N$))=Z$
290 C$(-1)=N$+C$(-1)
300 Rem LOOPS TO PRINT G GROUPS OF LABELS EACH WITH L LINES
310 Print
320   For J=1 To G
330     For K=1 To L
340       If K=1 Then Print A$
350       If K=2 Then Print B$
360       If K=3 Then Print C$
370     Next K
380   Next J
390 Print : Rem END GROUP OF LABELS WITH A BLANK LINE
400 Goto 120

```

```

>>run
WHAT REPLICATION FACTOR (EXIT WITH -1)  2
DATA FOR LINE 1  RIVERSIDE
DATA FOR LINE 2 (END LABEL WITH #)      CO
DATA FOR LINE 3 (END LABEL WITH #)      *

RIVERSIDE RIVERSIDE RIVERSIDE RIVERSIDE RIVERSIDE RIVERSIDE RIVERSIDE RIVERSIDE
  CO      CO      CO      CO      CO      CO      CO      CO
RIVERSIDE RIVERSIDE RIVERSIDE RIVERSIDE RIVERSIDE RIVERSIDE RIVERSIDE RIVERSIDE
  CO      CO      CO      CO      CO      CO      CO      CO

WHAT REPLICATION FACTOR (EXIT WITH -1)  -1
***130 Stop***
>>

```

Fig. 1. Listing of the program and an example of its use. Labels shown are actual size.

David G. Kissinger, *Department of Biostatistics and Epidemiology, School of Health, Loma Linda University, Loma Linda, California 92350.*

PROC. ENTOMOL. SOC. WASH.
84(4), 1982, pp. 857-858

NOTE

Ferrisia lobdellae, a New Name for *Ferrisia setosa*
(Lobdell, 1930) (Homoptera: Pseudococcidae)

Ferrisia setosa (Lobdell) is presently without a valid name. This mealybug was originally discovered on the roots of *Liquidambar styraciflua* at Durant, Mississippi, by Lobdell (1930. *Ann. Entomol. Soc. Am.* 23: 209-236) and named *Trionymus setosus*. Ferris (1950. *Atlas of the Scale Insects of North America, Series V: The Pseudococcidae, Pt. I, pp. 85-95*) treated it as a valid species and transferred it to the genus *Ferrisiana*. But the generic name *Ferrisiana* is now considered as an unjustified change for *Ferrisia* Fullaway, 1923 (*vide*, Morrison and Morrison, 1966. *U. S. Dep. Agric. Misc. Publ.* 1015: 1-206). Therefore, this species presently is called *Ferrisia setosa* (Lobdell).

However prior to Lobdell's 1930 paper, Hempel (1900. *Rev. Mus. Paulista, São Paulo* 4: 386) described a mealybug on *Ficus* in Brazil as *Dactylopius setosus*, and he mentioned it again in a second publication (Hempel, 1900. *Ann. Mag. Nat. Hist.* (7) 6: 396). Hempel's mealybug was referred to as *Pseudococcus setosus* (Hempel) by Fernald (1903. *Mass. Agric. Exp. Stn. Spec. Bull.* 88: 109) and by MacGillivray (1921. *The Coccidae*. Scarab Co., Urbana, Illinois, p. 133), was later placed in the genus *Ferrisia*, and